GRAPHICALLY CONQUERING THE SF-36: A TOOL FOR ILLUSTRATING SUBSCALE BY GROUP OVER TIME USING SAS/GRAPH® SOFTWARE

David Dorr, Mae Gordon, and the Ocular Hypertension Study (OHTS) group, Washington University School of Medicine

ABSTRACT

The widespread adoption of health related quality of life (QoL) surveys to measure patient outcomes in clinical practice as well as research has raised the issue of simple graphical display. Such surveys often have multiple subscales to quantify disparate dimensions in the QoL, including emotional, physical, and others. When this is combined with the need to present longitudinal data for multiple groups, concise and accessible summarization of results becomes difficult. This paper presents a straightforward way to graph subscales by group over time, using one of the most common surveys, the Short Form - 36. The method allows the all eight SF-36 subscales to be presented on a single sheet of paper, and, as we will show, highlights trends and changes both within and between groups.

NIH: EY09341, EY09307; Unrestricted grant from Research to Prevent Blindness

BACKGROUND

The MOS Short Form - 36 (SF-36) is a Quality of Life (QoL) tool developed to measure the impact of various medical diseases, conditions and the outcome of the treatments on a patient's life. Motivated by the need for a more sensitive tool for measuring patient outcome, the SF-36 evaluates various components of well-being, including mental health, physical health, vitality and the limitations these cause in the patient's life (1,2). Development of a new wave of surveys like the SF-36 gained impetus from a paradigm shift in health care in the United States. This shift focused on a person's health related QoL, both current and expected, as a measure for the relative effect of various illnesses as well as a metric for gauging the overall effect of treatment, rather than the traditional method of discrete medical outcomes.

The Ocular Hypertension Study (OHTS) is a prospective randomized controlled trial of clinical tests and efficacy measures in patients with ocular hypertension who are thought to be at higher risk for development of glaucoma. The treatment is the use of commercially available intraocular pressure lowering drugs; the major endpoint is development of open angle glaucoma. The SF-36 is used to both compare study participants to the norm of their age-matched peers, and to examine differences between groups. This comparison is made over various time intervals in order to better understand the longitudinal course and effect of the treatment and the endpoint in those who reach it. Multiple administrations provide a meter for non-pathological fluctuations in quality of life, allowing an increased sensitivity of the SF-36 surveys for whole study trends in QoL.

A difficulty posed by the SF-36 is the graphical illustration is the lack of a single summary index for the survey. Rather, multiple subscales which measure separate dimensions of a patient's quality of life are necessary for analysis and comparison of the overall quality of life. The current situation in OHTS added the necessity of presenting the various subscales (8 scales) by group (2+ groups) over time (up to 4-5 visits). This paper discusses a method to graphically display such a problem with clarity and concision.

METHODS

Since the OHT study is currently ongoing and the results are impounded, data were randomly generated to simulate OHTS results. The basic framework of the data set consisted of (1) the eight SF-36 scored scales (see *Step 1*, below) with a possible range of 0-100, but constrained in a normal distribution around the norms for US healthy 55-64 year olds; (2) a group variable, consisting of the medication and observation groups; and (3) a visit variable, with a baseline visit, and 4 6-mo. follow-ups. In a more general sense, any group and visit structure can be used, with the limitations largely extending from space and legibility constraints.

Our goal was to provide the reader with a visual summary of the results as well as specific numeric values ($\overline{X}$, SD) on a single page. To accomplish this, we examined several different classic graphical representations. The first was a bar chart (either horizontal or vertical) with scaled score on the vertical axis and scale by group by time on the horizontal axis or some permutation of the variable order. The difficulties immediately presented with this trial was: (1) illegibility and/or space limitation, (2) difficulty seeing trends, and (3) inability to compare at least one element of the comparison, either between groups or the different scales within groups. Next, we considered a connecting line plot; again, this was difficult to interpret. The difficulty here is also the density of the data, leading to either an oversimplification or a confusing result. The closest we came in SAS/GRAPH was using PROC GREPLAY to show four graphs presenting four visits simultaneously.

We decided to present the scale as a series of one dimensional comparisons between groups by plotting the group scores as points on a line (0-100). Then, we pushed the second dimension, the vertical, into two dimensions by using a proximal and distal component. The distal was the eight scales; the proximal was the visits requested for that scale. Essentially, this was concatenating the last example into a single graph.

The nature of such a graphic precluded a simple PROC GPLOT or other graphic procedure invocation, so we wrote the following code. The macro is flexible, to allow new and other visit dates to be included, as well as easy transition to other, similar outcome measures. In our experience, the transition to at least two different components of the study has been without difficulties.

Step 0: Macro variable definitions

GRAPHICALLY CONQUERING THE SF-36: A TOOL FOR ILLUSTRATING SUBSCALE BY GROUP OVER TIME USING SAS/GRAPH® SOFTWARE

David Dorr et al.

Variable	DS	BBY
Meaning	Dataset name	By variable
Example	*WORK.COOL*	*RA_BR*
Conditions	see Step 1	f=$1.
Variable	*VINO*	*VINAME*
Meaning	Number of vis	Visits
Example	5	00 06 12 18 24
Conditions		f=$2.

Step 1: Data set form

To use the macro as presented, the data must be in the following form:

```
VAR    VALUE  VISIT  By var
$1.    0-100    $2.    $1.
```

where VAR is a one letter code for the various scales of the survey, ordered appropriately, according to the following PROC FORMAT:

```
'1'="PHYSICAL"
'2'="ROLE-PHYSICAL"
'3'="BODILY PAIN"
'4'="HEALTH PERCEPT."
'5'="VITALITY"
'6'="SOCIAL FUNCTION"
'7'="ROLE-MH"
'8'="MENTAL HEALTH"
```

Step 2: Normals

Macro %QBS provides the norm mean and standard deviation for each visit. These are set internally by user; in our example, we use the SF-36 published norms for age 55-64.

Step 3: Set-up variables / Optional annotate

First, the macro completes the dataset by combining the variables VAR and VISIT. Then, an optional annotate is added. In our current example, the annotate provides the mean and standard deviation of each discrete value in the group variable. This requires two vertical columns per value, so it is not recommended for large numbers of different groups. The annotate variables *V1-V4* are positional variables. (V1,V2) are the X,Y starting coordinates of the columns. V3 is the distance between the BY variable groupings, and V4 is the distance between the Mean and Standard Deviation columns.

Step 4: Graphics set-up and PROC GPLOT

Finally, the program sets graphics options, including symbols, axes, and the legend. In order to have the maximum amount of control over presentation of the visit numbers, the TICK option is used in the vertical axis. This requires iterative %DO loops to account for changing number of visits (*see program)*, but it allows one to left justify the subscales while right justifying the visit names. To enhance this subscale separation effect, vertical reference lines are dynamically generated for the visit lines only.

```
*********************************************
```

```
( SF36G.SAS SF-36 TIME SERIES GRAPH            )
( ORIGINAL PROGRAM BY DAVID DORR, OHTS         )
( PURPOSE:  SUMMARIZE SF-36 OVER MULTIPLE TIME)
(   PERIODS BY GROUPS.  SAS GRAPH NEEDED.      )
*********************************************;
```

Step 0: Macro variable definitions

```
%MACRO SF36G(DS,BBY,VINO,VINAME,TITL,V1,V2,V3,V4);
```

Step 1: Data set form

```
* GENRAN;
%MACRO GENRAN(MNY,SDY);
DO I = 1 TO 1000;
IF I<=500 THEN RA_BR='1'; ELSE RA_BR='2';
   VALUE=(RANNOR(0) * &SDY. ) + &MNY.;
*** Deleted (dataset creation);
OUTPUT;
END;
%MEND GENRAN;

%MACRO QL_MN(VI);
DATA MO&VI.; SET &DS.; LENGTH VAR $30 VISIT $2;
   VALALL='3';
VAR="1"; VISIT="&VI."; %GENRAN(73.3, 26.3);
*..... etc;
RUN;
DATA MO&VI.; SET MO&VI.; KEEP VAR VISIT VALUE &BBY.;
RUN;
%MEND;
**********************************************
* LOAD VISIT NAMES INTO MACRO VARIABLES
**********************************************;
DATA BLANK; LENGTH TVAR $30 VS1-VS&VINO. $2 C1 C2 $1;
ARRAY DU{2} $ C1 C2;
ARRAY VI{*} $ VS1-VS&VINO.;
TVAR = "&VINAME."; N=0; J=1;
DO I = 1 TO LENGTH(TVAR);
   TL= SUBSTR(TVAR,I,1);
   IF TL NE ' ' THEN DO;
    DU{J}=TL; IF J=1 THEN J=2;
    ELSE DO;
     J=1; N+1;
     VI{N} = COMPRESS(C1||C2);
     FLAV= COMPRESS("VS"||N);
     CALL SYMPUT (FLAV, VI{N});
   END;
  END;
END;
RUN;
*************************************
* MUST SET W/ NECESSARY VISIT NUMBERS*
*************************************;
%DO J = 1 %TO &VINO.;
%QL_MN(&&VS&J);
%END;
```

Step 2: Normals

```
%MACRO QBS(VI,BYVAR);
VISIT="&VI."; &BYVAR.="0";
 VAR='1'; VALU2=76.3; VALU3=26.3; OUTPUT;
* . . . . etc (1-8);
%MEND QBS;
**********************************************
* ALSO MUST SET W/ NECESSARY VISIT NUMBERS.   *
* HERE, !! IS PLACEHOLDER (SEE GRAPH)         *
**********************************************;
DATA NOR; LENGTH VAR $30 VISIT $30;
%QBS(!!,&BBY.);
%DO J = 1 %TO &VINO.;
%QBS(&&VS&J,&BBY.);
%END;
RUN;

***********************************
*LOAD THE VISITS YOU WANT         *
***********************************;
```

Step 3: Set-up variables / Optional annotate

```
DATA NEW; SET
%DO J = 1 %TO &VINO.;
 MO&&VS&J
%END; ;
```

GRAPHICALLY CONQUERING THE SF-36: A TOOL FOR ILLUSTRATING SUBSCALE BY GROUP OVER TIME USING SAS/GRAPH® SOFTWARE

David Dorr et al.

```
NVAR=TRIM(VAR) || TRIM(VISIT);
RUN;
DATA UNEW;
DO I = 1 TO 8;
  VAR=PUT(I,$1.); VISIT='!!'; VALUE=.; RA_BR='1';
  VALALL='3';
  NVAR=TRIM(VAR) || TRIM(VISIT); OUTPUT;
END;
RUN;
DATA NOR; SET NOR;
  NVAR=TRIM(VAR) || TRIM(VISIT);
RUN;
DATA NEW; SET NEW UNEW; RUN;
PROC SORT DATA=NEW; BY &BBY. VISIT NVAR;
PROC MEANS DATA=NEW NOPRINT N MEAN STD;
BY &BBY. VISIT NVAR; VAR VALUE;
OUTPUT OUT=NEW2U MEAN=VALU2 STD=VALU3;
RUN;
DATA NEW2U; SET NOR NEW2U;

************************************************
* ANNOTATE FOR RIGHT HAND SIDE COLUMNS MN STD *
***********************************************;

DATA ANNO; SET NEW2U; FUNCTION='LABEL'; BY &BBY. VISIT;
IF FIRST.VISIT THEN ONCE=0; ELSE ONCE=1;
 IF ONCE NE 1 THEN ONCE=0;
LENGTH TEXT $20 YC $6;
IF ONCE NE 1 THEN DO;
XSYS='5'; YSYS='5'; POSITION='5';
SIZE=.7;
  X=&V1.; Y=&V2.; TEXT='Norm'; OUTPUT;
  X=(&V1.-&V4.); Y=(&V2.-2); TEXT='Mean'; OUTPUT;
  X=(&V1.+&V4.); Y=(&V2.-2); TEXT='(SD)'; OUTPUT;
%IF "&BBY." NE "VALALL" %THEN %DO;
  X=(&V1.+&V3.); Y=&V2.; TEXT='Med'; OUTPUT;
  X=(&V1.+&V3. -&V4.); Y=(&V2.-2); TEXT='Mean'; OUTPUT;
  X=(&V1.+&V3. +&V4.); Y=(&V2.-2); TEXT='(SD)'; OUTPUT;
  X=(&V1.+(2* &V3.)); Y=(&V2.); TEXT='Obs'; OUTPUT;
  X=(&V1.+(2* &V3.) -&V4.); Y=(&V2.-2); TEXT='Mean';
OUTPUT;
  X=(&V1.+(2* &V3.) +&V4.); Y=(&V2.-2); TEXT='(SD)';
OUTPUT;
%END;
%ELSE %DO;
  X=(&V1.+&V3.); Y=&V2.; TEXT='All'; OUTPUT;
  X=(&V1.+&V3. -&V4.); Y=(&V2.-2); TEXT='Mean'; OUTPUT;
  X=(&V1.+&V3. +&V4.); Y=(&V2.-2); TEXT='SD'; OUTPUT;
%END;
ONCE=1;
END;
SIZE=.6;
XSYS='5'; YSYS='2'; POSITION='5';
%LET TX=;
   YC=NVAR; X=&V1. - &V4.; TEXT=PUT(VALU2,4.1);
  IF SUBSTR(&BBY.,1,1)='0' AND VISIT='!!'  THEN DO;
OUTPUT; END;
   YC=NVAR; X=&V1. + &V4.;
    TEXT='('||PUT(VALU3,4.1)||')';
  IF SUBSTR(&BBY.,1,1)='0' AND VISIT='!!' THEN DO;
OUTPUT; END;
   YC=NVAR; X=(&V1.+ &V3. - &V4.); TEXT=PUT(VALU2,4.1);
  IF  SUBSTR(&BBY.,1,1)='3' AND VISIT NE '!!' THEN DO;
OUTPUT; END;
   YC=NVAR; X=(&V1. + &V3. + &V4.);
    TEXT='('||PUT(VALU3,4.1)||')';
  IF SUBSTR(&BBY.,1,1)='3' AND VISIT NE '!!' THEN DO;
OUTPUT; END;
   YC=NVAR; X=(&V1. + &V3. - &V4.);
TEXT=PUT(VALU2,4.1);
  IF SUBSTR(&BBY.,1,1)='1' AND VISIT NE '!!' THEN DO;
OUTPUT; END;
   YC=NVAR; X=(&V1. + &V3. + &V4.);
    TEXT='('||PUT(VALU3,4.1)||')';
  IF SUBSTR(&BBY.,1,1)='1' AND VISIT NE '!!' THEN DO;
OUTPUT; END;
   YC=NVAR; X=(&V1. + (2 * &V3.) - &V4.);
TEXT=PUT(VALU2,4.1);
  IF SUBSTR(&BBY.,1,1)='2' AND VISIT NE '!!' THEN DO;
OUTPUT; END;
   YC=NVAR; X=(&V1. + (2 * &V3.) + &V4.);
    TEXT='('||PUT(VALU3,4.1)||')';
  IF  SUBSTR(&BBY.,1,1)='2' AND VISIT NE '!!' THEN DO;
OUTPUT; END;
PUT TEXT;
IF TEXT='.' THEN DELETE;
RUN;
   *PROC PRINT DATA=ANNO; RUN;
PROC PRINT DATA=NEW2U; RUN;
```

Step 4: Graphics set-up and PROC GPLOT

```
GOPTIONS RESET=ALL;
GOPTIONS DEVICE=xxx GUNIT=PCT ROTATE=LANDSCAPE
FTEXT=SWISSL  HTEXT=1.5;
SYMBOL1 C=BLACK I=NONE V=DOT H=1.5 WIDTH=5
   CV=BLACK MODE=EXCLUDE;
* . . . etc;
TITLE1 H=3 "Mean of Eight SF-36 Scales: Random
   Numbers   N=1000";
TITLE2 H=2.5 "&TITL. |  Includes normals for US
   population age 55-64";
FOOTNOTE1 H=1 "DD for SUGI23   Creation: &SYSDATE
&SYSTIME     DATASET: &DS.";
FOOTNOTE2 H=1 "~HOME/SF36G.SAS";
LEGEND1 VALUE=(H=1.5) LABEL=(H=1.5 'Legend')
    FRAME;
AXIS1 LABEL=(H=3 'SCORE') WIDTH=3
    MINOR=NONE VALUE=(H=3) LENGTH=54PCT
    ORDER=(0 20 40 60 80 100);
AXIS2 LABEL=(H=1.5 " ") WIDTH=3 OFFSET=( ,4PCT)
 VALUE=(H=1.2
%LET VAR1 =Physical; %LET VAR2=Role-Physical;
%LET VAR3 =Bodily Pain;
%LET VAR4=Health Percept.;
%LET VAR5 =Vitality; %LET VAR6=Social Function;
%LET VAR7 =Role-MH; %LET VAR8 =Mental Health;
%LET CNT=0;
%DO I = 1 %TO 8;
 %LET CNT=%EVAL(&CNT. +1);
 TICK=&CNT. J=L COLOR=BLACK "&&VAR&I"
 %DO J = 1 %TO &VINO.;
  %LET CNT=%EVAL(&CNT. +1);
  TICK=&CNT. J=R COLOR=BLACK "&&VS&J"
 %END;
%END;
);
PROC GPLOT DATA=NEW2U ANNOTATE=ANNO;
  PLOT NVAR*VALU2=&BBY./VAXIS=AXIS2
  HAXIS=AXIS1 VREF=
        %DO LP1 = 1 %TO 8;
          %DO LP2 = 1 %TO &VINO.;
           "&LP1&&VS&LP2"
        %END; %END;
  VREVERSE  LEGEND=LEGEND1;
FORMAT VAR $SFLONG.
  RA_BR VALALL $G2P.;
        RUN;
        QUIT;
%MEND SF36G;

%SF36G(NOTHING,VALALL,5,00 01 02 03
    04,All,78,98,8,1.8);
%SF36G(NOTHING,RA_BR,4,00 01 02 03,By Rand
    Group,78,98,6.5,1.5);
```

GRAPHICALLY CONQUERING THE SF-36: A TOOL FOR ILLUSTRATING SUBSCALE BY GROUP OVER TIME USING SAS/GRAPH® SOFTWARE

David Dorr et al.

Figure 1



Figure 2



GRAPHICALLY CONQUERING THE SF-36: A TOOL FOR ILLUSTRATING SUBSCALE BY GROUP OVER TIME USING SAS/GRAPH® SOFTWARE

David Dorr et al.

RESULTS

The resulting graphs are shown on the previous page, one with only one subgroup and the norms and one with two subgroups. In this case, we attempted to focus on clarity, simplicity, and ease of trend recognition for a historically complex graphic. We believe this presentation's simplicity facilitates analysis while still providing hard data.

The overall effectiveness of the graphic is left to the viewer. In our synthetic data, we have added multiple such trends of varying effect size and proportion. An excellent test of our methodology is to attempt to identify and characterise these trends. The answers are given in the following reference (3).

References

(1) Stansfeld SA. Roberts R. Foot SP. Assessing the validity of the SF-36 General Health Survey.
Quality of Life Research. 6(3):217-24, 1997 Apr.

(2) Anderson RT. Aaronson NK. Wilkin D. Critical review of the international assessments of health-related quality of life.
Quality of Life Research. 2(6):369-95, 1993 Dec.

(3) Health Perceptions +3(visit),
Mental Health -1(visit),
Role Physical 1.8^{Visit}